

Symulacja układów elektronicznych z użyciem oprogramowania SPICE zajęcia warsztatowe SKN CHIP

Przygotował

Bogdan Pankiewicz, maj 2017

Symulacja układów z użyciem SPICE – zajęcia SKN CHIP

Plan zajęć:

- Krótkie wprowadzenie historyczne.
- Właściwości ogólne symulatorów.
- Symulacja: oczekiwane rezultaty oraz sposób przeprowadzenia symulacji.
- Podsumowanie właściwości poszczególnych typów symulacji.
- Prosty przykład symulacji: równoległy obwód RLC z diodą półprzewodnikową. Symulacja z poziomu pliku tekstowego oraz z poziomu schematu.
- Przykład symulacyjny II: wzmacniacz tranzystorowy WE.
- Przykład symulacyjny III: przerzutnik Schmitta z wykorzystaniem WO.

Literatura

- Materiały do przedmiotu JMiS:
http://www.ue.eti.pg.gda.pl/~bpa/jmis/slajdy_jmis_1v3.pdf
- J. Izydorczyk, „PSpice komputerowa symulacja układów elektronicznych”, Helion, 1993.
- Cadence Design Systems, „PSPICE reference guide”, Cadence PCB Design Systems, 2001.
Dokumentacja w wersji elektronicznej dostępna w laboratorium w katalogu
C:\Cadence\PSD_14.0\doc.

SPICE, PSPICE, LTSPICE i inne symulatory

Krótką historia:

- SPICE – ang. Simulation Program with Integrated Circuit Emphasis, ze względu na koszty produkcji układów scalonych bardzo ważną stała się ich wstępna symulacja, jeszcze przed procesem produkcji i stąd powstała potrzeba symulacji.
- Pierwsza wersja symulatora powstaje w 1973r na University of California, Berkeley i jest sponsorowana przez United States Department of Defense.
- Następne wersje są kontynuacją, powstają również wersje komercyjne, SPICE natomiast staje się oprogramowaniem typu Public Domain.
- Obecnie mamy dostępne również inne symulatory obwodów elektrycznych: Eldo, Spectre, Hspice, PSpice, Saber i inne.
- PSPICE - Personal Simulation Program with Integrated Circuit Emphasis, wytworzony inicjalnie przez firmę MicroSim, następnie wykupiony przez OrCad a następnie przez Cadence Design Systems.
- Ltspice – symulator darmowy i w „pełnej” wersji, dostępny tu: <http://www.linear.com/designtools/software/>

Symulatory – właściwości ogólne

- Rodzaje wykonywanych symulacji – główne typy:
 - stałoprądowa (DC),
 - zmiennoprądowa małosygnałowa (częstotliwościowa) (AC),
 - czasowa (TRAN).
- Oprócz głównych 3 typów każdy z symulatorów dysponuje szeregiem symulacji pochodnych bazujących na analizach podstawowych:
 - DC $\rightarrow$ TF,
 - AC $\rightarrow$ NOISE,
 - TRAN $\rightarrow$ FOUR.
- Wejściem dla symulatora jest plik tekstowy zawierający:
 - tytuł badanego obwodu,
 - listę połączeniową,
 - modele elementów,
 - polecenia symulacyjne.

Symulatory – właściwości ogólne c.d.

Składnie pliku wejściowego:

- typu SPICE wraz z drobnymi odmianami,
- typu SPECTRE (symulator Spectre firmy Cadence, który również wczytuje format SPICE),
- inne.

Graficzna forma wprowadzania schematu badanego układu – właściwości:

- brak ogólnej standaryzacji pomiędzy różnymi programami,
- potrzeba zdefiniowania symulacji, modeli, opcji, i.t.d.,
- forma graficzna i tak generuje plik tekstowy np. w formacie SPICE,
- symulacja jest i tak przeprowadzana na wygenerowanym pliku tekstowym.

Oczekiwane rezultaty

- Obliczenie wartości występujących w obwodzie napięć węzłowych i prądów gałęziowych:
 - napięcia i prądy stałe (DC),
 - amplitudy i fazy przebiegów harmoniczných w stanie ustalonym dla pobudzeń harmoniczných na zlinearyzowanym modelu układu (AC),
 - wartości napięć i prądów w funkcji czasu (TRAN).
- Na podstawie w.w. wartości wyliczenie parametrów układu takich jak np.:
wzmocnienie, amplitudy harmoniczných, szum całkowity, THD, DR, itd.

Sposób przeprowadzenia symulacji.

- Przygotowanie/ściągnięcie modeli elementów wykorzystywanych w badanym układzie.
- Wprowadzenie schematu badanego obwodu.
- Wprowadzenie niezbędnych typów analiz.
- Wykonanie symulacji.
- Odczyt wyników symulacji.
- Ewentualne przetworzenie wyników symulacji w celu uzyskania dodatkowych informacji o badanym układzie.

Informacje te mogą być wprowadzone
za pomocą pliku tekstowego lub schematu oraz szeregu menu.

Ogólna składnia opisu SPICE

Układ 1

R1 1 0 10k

C1 1 0 1f

* to jest komentarz

.op

.end

Pierwsza linia - tytuł obwodu, dozwolona dowolna zawartość tej linii.

Kolejne linie zaczynające się od litery – lista połączeniowa obwodu.

Linie zaczynające się „*” – komentarz trwający do końca linii.

Linie zaczynające się kropką stanowią polecenia symulatora.

Ogólna składnia opisu SPICE – c.d.

- typowe rozszerzenie nazwy pliku to „**.cir**”, inne rozszerzenia mogą powodować problemy w działaniu oprogramowania,
- należy unikać polskich znaków i spacji zarówno w nazwach plików jak i w opisie wewnątrz pliku SPICE,
- składnia SPICE jest nieczuła na wielkość liter,
- w pierwszej linii podaje się zwięzły opis badanego układu,
- ostatnia linia pliku powinna zawierać polecenie **.end**, jeśli plik zawiera linie za **.end** traktowane są one jako kolejny obwód do analizy,
- puste linie są pomijane,

Ogólna składnia opisu SPICE – c.d.

- linie rozpoczynające się literą są opisem elementu zależnego od tego jaka jest to litera np.: R- rezystor, L- cewka, C- kondensator i.t.d.,
- linie rozpoczynające się kropką są poleceniami, parametrami lub dodatkowym sterowaniem pracą symulatora,
- linie rozpoczynające się gwiazdką „*” są komentarzem trwającym do końca linii,
- linie w których występuje średnik ; – od pozycji średnika do końca linii też są traktowane jako komentarz,

Ogólna składnia opisu SPICE – c.d.

- długie linie (typowo ponad 80 znaków) można przenieść do kolejnej linii opisu rozpoczynając ją znakiem „+”

Przykład:

R1 1 0 22k

można zapisać:

R1 1 0

+ 22k

Ogólna składnia opisu SPICE – c.d.

- węzeł masy ma w SPICE zawsze nazwę „0” (zero),
- węzeł ten **musi** wystąpić w badanym układzie,
- każdy z węzłów musi mieć przynajmniej 2 połączenia do elementów – w przeciwnym przypadku węzeł będzie typu „floating” i zablokuje możliwość symulacji,
- niektóre symulatory samodzielnie usuwają węzły „floating” i umożliwiają symulacje generując jedynie ostrzeżenie o potencjalnym problemie,

Ogólna składnia opisu SPICE – c.d.

- również węzły, które mają 2 połączenia ale wyłącznie zrealizowane poprzez kondensatory są typu „floating”,
- niedozwolone jest równoległe łączenie idealnych źródeł napięciowych,
- niedozwolone jest szeregowe łączenie idealnych źródeł prądowych.

Ogólna składnia opisu SPICE – c.d.

- Większość używanych jednostek to jednostki S.I.
(tylko pojedyncze przypadki w niektórych modelach elementów są spoza układu S.I.)
- można stosować przyrostki wartości:

f	p	n	u	m	k	meg	g	t	mil
F	P	N	U	M	K	MEG	G	T	MIL
10^{-15}	10^{-12}	10^{-9}	10^{-6}	10^{-3}	10^3	10^6	10^9	10^{12}	$25.4 \cdot 10^{-6}$

Ogólna składnia opisu SPICE – c.d.

Formaty podawania wartości parametrów elementów:

- zwykły, np.: **123.23**, - punktem dziesiętnym jest kropka, można nie podawać zera przed wartością np.
0.123=.123,
- naukowy, np.: **1.2323e-5**,
- z użyciem przyrostka, np.: **1.2323m**,
- z dodatkowym podaniem jednostki, np.: **1.2323e-3V** lub **1.2323mV**,
- uwaga na jednostkę F gdyż jest to przyrostek oznaczający 10^{-15} więc:
1.2323e-6F nie jest równe 1.2323uF

Ogólna składnia opisu SPICE – c.d.

- wartości można również podawać pośrednio poprzez wykorzystanie wyrażenia umieszczonego w nawiasie klamrowym, np.: `R1 1 0 {10k*2/23}`,
- wartości wyliczane w nawiasie klamrowym mogą wykorzystywać bardziej złożone zależności poprzez zastosowanie operatorów, funkcji oraz możliwość umieszczenia w wyrażeniu parametrów, np.:

```
.param mult=10
```

```
R1 1 0 {mult*2*sin(mult)*1K}
```

Analiza stałoprądowa (DC)

Analiza **.DC** wykonuje sparametryzowane obliczenie punktu pracy układu. Elementy inercyjne są pomijane (C -> rozwarcie, L -> zwarcie), elementy nieliniowe zachowują nieliniowy charakter. Argumentem tej analizy jest zmieniany parametr (pierwszy lub oba) a wynikiem rozptyw stałych prądów w układzie oraz wartości stałych napięć w węzłach.

Analiza **.DC** występuje w 4 formach przemiatania parametrów analizy: **LIN**, **OCT**, **DEC** oraz **LIST**. Analiza może być zagnieżdżona tzn. zmiana parametru pierwszego jest wykonywana w pętli wewnętrznej i następuje częściej niż parametru drugiego.

Analiza stałoprądowa (DC) - c.d.

Format ogólny dla przemiatań liniowego (LIN):

`.DC [LIN] <nazwa przemiataanej zmiennej>`

`+ <start> <koniec> <wartość inkrementu>`

`+ [specyfikacja zagnieżdżenia]` - (takie same zasady jak dla podstawowego przemiatań)

Przykłady:

`.DC VIN -.25 .25 .05`

`.DC LIN I2 5mA -2mA 0.1mA`

`.DC VCE 0V 10V .5V IB 0mA 1mA 50uA`

Analiza małosygnałowa (AC)

Format ogólny:

```
.AC <typ przemiatania> <liczba punktów>  
+ <częstotliwość początkowa>  
+ <częstotliwość końcowa>
```

Przykłady:

```
.ac lin 201 1k 2k - 201 punktów częstotliwościowych
```

```
.ac dec 50 100Hz 1GHz - 50 punktów na każdą dekadę
```

```
.ac oct 30 1 20 - 30 punktów na każdą oktawę
```

Analiza małosygnałowa (AC) - c.d.

Właściwości analizy:

- jest to analiza liniowa,
- najpierw obliczany jest punkt pracy,
- następnie linearyzowane są wszystkie elementy nieliniowe (zastępowane modelami małosygnałowymi),
- następnie obliczana jest charakterystyka częstotliwościowa,
- wejściami do układu są wszystkie niezależne źródła napięciowe i prądowe z niezerową specyfikacją AC
- wszystkie obliczone wielkości są liczbami zespolonymi.

Analiza czasowa (TRAN)

Format ogólny:

```
.TRAN [/OP] <krok_wydruku> <czas_analizy>  
+[nie_drukować [krok_obliczeniowy]] [SKIPBP]
```

Przykłady:

.TRAN 1ns 100ns - analiza do 100ns z krokiem wydruku 1ns

.TRAN 1ns 100ns 0ns .1ns SKIPBP – analiza do 100ns z krokiem obliczeniowym 0,1ns bez obliczania punktu pracy

Opcja **/OP** powoduje umieszczenie szczegółowych danych dotyczących punktu pracy w pliku wyjściowym „***.out**”.

Opcja **SKIPBP** powoduje pominięcie obliczania punktu pracy, gdy użyte punktu pracy jest ustalony za pomocą wartości **IC** w deklaracjach kondensatorów i cewkach.

Analiza czasowa (TRAN) – c.d.

Właściwości analizy .TRAN:

- oblicza zachowanie się układu w czasie, argumentem dla analizy jest czas,
- analiza zawsze rozpoczyna się w czasie równym 0,
- analiza kończy się w czasie `czas_analizy`,
- przed rozpoczęciem analizy obliczany jest punkt pracy, który może być inny niż dla AC bo mogą być inne wartości specyfikacji pobudzeń,

Analiza czasowa (TRAN) – c.d.

- domyślny krok analizy dobierany jest w zależności od aktywności układu (zmian w układzie),
- domyślna wartość kroku analizy wynosi **czas_analizy/50**, ale w przypadku gdy nie ma w układzie elementów inercyjnych krok analizy jest równy **krok_wydruku**,
- analiza **.TRAN** ustawia wartości globalnych zmiennych **TSTEP** oraz **TSTOP**, które to są następnie używane do wyznaczania niektórych wartości domyślnych,
- **TSTEP = krok_wydruku**
- **TSTOP = czas_analizy**


Analiza czasowa (TRAN) – c.d.

- dla parametru `krok_obliczeniowy` można zastosować funkcję:
`SCHEDULE(time1, val1, time2, valn...timen, valn),`
- funkcja ta określa jaki krok obliczeniowy ma być użyty w kolejnych przedziałach czasowych,
- znaczenie jest następujące: od czasu `time1` użyj wartości `val1`, następnie od czasu `time2` użyj wartości `val2` i.t.d.

Przykład użycia funkcji:

```
.TRAN 1ns 90ns 0ns  
+ { SCHEDULE(0, 1ns, 25ns, .1ns) }
```

Przykład symulacji I: obwód rezonansowy RLC z diodą półprzewodnikową



ex_1

obwód rezonansowy

```
L1 2 0 100uH
```

R11 1 2 10

```
C1 1 0 10nF
```

R1 1 0 1k

D1 1 0 D1n4007

```
.lib ln4007.lib
```

*źródło pradowe

```
I1 0 1 dc 0 ac 1 pulse(0 10 5u 5n 5n .1u 1)
```

```
*.ac dec 100 1 10meg
```


```
*.dc i1 0 10m .01m
```

```
.tran 0 .1m 0 .1u
```

```
.probe
```

```
.end
```

Przykład symulacji II: kaskoda BJT




```
Ex No 2 - BJT cascade
Q1 c1 b1 e1 0 BC238
Q2 c2 b2 c1 0 BC238
R1 vdd b2 4k
R2 b2 b1 2k
R3 b1 0 4k
R4 e1 0 3.3k
R5 vdd c2 2k
R6 out 0 2k
R7 b1 N001 100
V1 in 0 dc 0 ac 1 SINE(0 10mV 10k)
Vdd vdd 0 10V
C1 1 in 1u
C2 e1 0 100u
C3 out c2 1u
C4 b2 0 100u
.inc bc238.txt
.op 0 1m 0 1u
;ac dec 50 1 1G
;tran 0 1m 0 1u
.probe
.end
```

Podukłady – na przykładzie wstawienia bramki NAND

Przykład – 2 x bramka NAND

Opis bramki umieszczony pomiędzy słowami kluczowymi `.sub` i `.ends`.

```
*Zdefiniowanie bramki NAND
*   NAZWA      węzły połączeniowe
*   |          |
.sub nand      in1 in2 out vdd
M1  10 in1    0   0  nfet W=0.9u L=0.6u
M2  out in2   10   0  nfet W=0.9u L=0.6u
M3  out in1  vdd vdd pfet W=0.9u L=0.6u
M4  out in2  vdd vdd pfet W=0.9u L=0.6u
.ends
```



Węzeł wewnętrzny nazwany „10”


Podukłady – na przykładzie wstawienia bramki NAND - c.d.

```


*Zasilanie
Vdd vdd 0 3
*Sygnał wejściowy
Vin w1 0 dc 0 pulse(0 3 1n 1p 1p 1n 2n )
* Badany układ 2 bramek jako inwerterów
* węzły połączeniowe      NAZWA
* | | | | | | | | | | | | | |
X1 w1      w1      w2      vdd      nand
X2 w2      w2      w3      vdd      nand
*Zdefiniowanie bramki NAND
*      NAZWA      węzły połączeniowe
*      |          | | | | | | | | | | | |
.sub nand          in1 in2 out vdd
M1 10 in1      0      0 nfet W=0.9u L=0.6u
M2 out in2     10      0 nfet W=0.9u L=0.6u
M3 out in1     vdd     vdd pfet W=0.9u L=0.6u
M4 out in2     vdd     vdd pfet W=0.9u L=0.6u
.ends
*Analizy
.tran 5p 5n 0n 5p
.probe
.inc scn05mos59jmagic.lib
.end

```


Pełny plik symulacyjny



Przykład symulacji III: przerzutnik Schmitta (rys zaczerpnięty z Wiki)



```
EX_3
Schmitt trigger
.inc tl071.301
X1 1 in vdd vss out tl071
R1 1 0 1k
R2 out 1 1k
Vdd vdd 0 12
Vss vss 0 -12
Vin in 0 dc 0 AC 1 pulse(-10 10 1m 1m 1m 1u 2.002m)
.dc vin -10 10 .1
.tran 5m
.probe
```



Przykład symulacji III: przerzutnik Schmitta (rys zaczerpnięty z Wiki)

